/*
* Licensed under the MIT License
* 
* Copyright (c) 2010 Kay Kreuning
* 
* Permission is hereby granted, free of charge, to any person obtaining a copy
* of this software and associated documentation files (the "Software"), to deal
* in the Software without restriction, including without limitation the rights
* to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
* copies of the Software, and to permit persons to whom the Software is
* furnished to do so, subject to the following conditions:
* 
* The above copyright notice and this permission notice shall be included in
* all copies or substantial portions of the Software.
* 
* THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
* IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
* FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
* AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
* LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
* OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
* THE SOFTWARE.
* 
* http://code.google.com/p/as3hyvesapi
*/
package nl.hyves.api.sessions
{
	import nl.hyves.api.HyvesService;
	import nl.hyves.api.errors.HyvesError;
	import nl.hyves.api.methods.PrivateMethods;
	import nl.hyves.api.utils.AuthUtil;
	import nl.hyves.api.utils.TokenStore;
	
	public class LogintokenSession implements ISession
	{
		protected var logintoken:String;
		protected var forceNewToken:Boolean;
		protected var serviceId:String;
		protected var callback:Function;
		
		public function LogintokenSession(logintoken:String, forceNewToken:Boolean = false)
		{
			this.logintoken = logintoken;
			this.forceNewToken = forceNewToken;
		}
		
		/**
		 * @inheritDoc
		 */
		public function connect(service:HyvesService, callback:Function):void
		{
			serviceId = service.id;
			
			if (TokenStore.has(serviceId) && !forceNewToken)
			{
				callback(TokenStore.get(serviceId), null);
			}
			else
			{
				service.call(PrivateMethods.auth_accesstokenByLogintoken, onAccesstoken, {logintoken: logintoken});
				
				this.callback = callback;
			}
		}
		
		private function onAccesstoken(result:Object, error:HyvesError):void
		{
			if (error)
			{
				callback(null, error);
				return;
			}
			
			var token:OAuthToken = AuthUtil.extractToken(OAuthToken.ACCESSTOKEN, result);
			
			TokenStore.set(serviceId, token);
			
			callback(token, null);
		}
	}
}